
Wren Hollow Church sample. Giving Analytics Dashboard Power BI rebuild guide

Giving Analytics: Power BI rebuild guide
This guide lists everything needed to rebuild the Giving Analytics Dashboard in Power BI
Desktop: the data files, the relationships, the calculated columns, every measure in DAX, and
the visuals on each page. The interactive HTML version in this kit shows what the finished report
should look like.

Files in this kit

File What it is

data/Giving Analytics - Data
Tables.xlsx

One Excel table per sheet: Donations, Designations, Funds, Batches, Pledges,
Pledge Campaigns, General Fund Budget, People, Households. ID columns are
stored as text.

data/*.csv The same tables as CSV files, plus the Check-Ins, Groups, Services, and Workflows
tables used by the other reports in the suite.

rebuild/Wren Hollow Giving
Theme.json

Power BI report theme (colors, fonts, card style). Import it under View, Themes,
Browse for themes.

dashboard/Giving Analytics
Dashboard.html

The finished layout to match.

The tables are shaped like Planning Center Giving and People data, and the column names follow
Planning Center's API field names (for example amount_cents, received_at, payment_method, and

membership). In a live setup these files are replaced by the church's own exports or by a scheduled pull

from the Planning Center API, and the model below stays the same.

Step 1. Load and type the data

1. Get Data, Excel workbook, and select all nine tables from the workbook.

2. In Power Query, set every column ending in _id (and People[id]) to Text so IDs are never summed.

3. Set Donations[received_at] and the other date columns to Date (or Date/Time for created_at fields),

amounts in cents to Whole Number, and Donations[refunded] and Donations[fee_covered] to
True/False.

4. Set 'General Fund Budget'[month_start] to Date and [budget_amount] to Fixed Decimal.



Step 2. Relationships

From (many side) To (one side) Notes

Designations[donation_id] Donations[donation_id] A gift can be split across funds, so
amounts live on Designations.

Designations[fund_id] Funds[fund_id]

Donations[person_id] People[id] Anonymous plate cash has no person.

People[household_id] Households[household_id]

Donations[received_at] 'Date'[Date] Mark 'Date' as the date table.

'General Fund
Budget'[month_start]

'Date'[Date] Budget is monthly, keyed to the first
day of each month.

Pledges[person_id] People[id]

Pledges[pledge_campaign_id] 'Pledge
Campaigns'[pledge_campaign_id]

Keep every relationship single direction. Measures that count giving units use the Designations table as
a filter argument, which lets a fund filter reach the Donations table without turning on two-way filtering.

Step 3. Calculated columns, tables, and measures

Create a blank table named _Measures to hold the measures. Formatting: currency with no decimals for
dollar measures, whole numbers for counts, and percentage with one decimal for shares.

Calculated columns

Designations[Amount] =

Designations[amount_cents] / 100

Donations[Is Recurring] =

LEN ( Donations[recurring_donation_id] ) > 0

Donations[Giving Unit] =

VAR hh = RELATED ( People[household_id] )

RETURN

    IF (

        LEN ( Donations[person_id] ) = 0,

        BLANK (),

        IF ( LEN ( hh ) = 0, "P" & Donations[person_id], "H" & hh )

    )

Pledges[Giving Unit] =

VAR hh = RELATED ( People[household_id] )



RETURN IF ( LEN ( hh ) = 0, "P" & Pledges[person_id], "H" & hh )

Pledges[Received] =

VAR u = Pledges[Giving Unit]

RETURN

    CALCULATE (

        SUM ( Designations[Amount] ),

        REMOVEFILTERS (),

        Funds[name] = "Room to Grow",

        Donations[refunded] = FALSE (),

        Donations[Giving Unit] = u

    )

Pledges[Expected to Date] =

VAR asOf = CALCULATE ( MAX ( Donations[received_at] ), REMOVEFILTERS () )

VAR monthsIn = DATEDIFF ( DATE ( 2025, 10, 1 ), asOf, DAY ) / 30.4375

RETURN Pledges[amount_cents] / 100 * MIN ( 1, monthsIn / 36 )

Pledges[Pace Status] =

VAR r = DIVIDE ( Pledges[Received], Pledges[Expected to Date] )

RETURN

    SWITCH (

        TRUE (),

        Pledges[Received] <= 0, "Not started",

        r >= 1.1, "Ahead of pace",

        r >= 0.85, "On pace",

        "Behind pace"

    )

Calculated tables

Date =

VAR firstDay = DATE ( YEAR ( MIN ( Donations[received_at] ) ), 1, 1 )

VAR lastDay = DATE ( YEAR ( MAX ( Donations[received_at] ) ), 12, 31 )

RETURN

    ADDCOLUMNS (

        CALENDAR ( firstDay, lastDay ),

        "Year", YEAR ( [Date] ),

        "Month Number", MONTH ( [Date] ),

        "Month", FORMAT ( [Date], "mmm yyyy" ),

        "Month Start", DATE ( YEAR ( [Date] ), MONTH ( [Date] ), 1 )

    )

Giving Bands =

DATATABLE (

    "Band", STRING, "Min", CURRENCY, "Max", CURRENCY, "Sort", INTEGER,

    {

        { "Under $250", 0, 250, 1 },

        { "$250 to $999", 250, 1000, 2 },

        { "$1,000 to $2,499", 1000, 2500, 3 },

        { "$2,500 to $4,999", 2500, 5000, 4 },



        { "$5,000 to $9,999", 5000, 10000, 5 },

        { "$10,000 to $24,999", 10000, 25000, 6 },

        { "$25,000 and up", 25000, 999999999, 7 }

    }

)

Measures (store them in a table named _Measures)

Report Date =

CALCULATE ( MAX ( Donations[received_at] ), REMOVEFILTERS () )

Total Given =

CALCULATE ( SUM ( Designations[Amount] ), Donations[refunded] = FALSE () )

General Fund Given =

CALCULATE ( [Total Given], Funds[name] = "General Fund" )

Giving Units =

CALCULATE (

    DISTINCTCOUNT ( Donations[Giving Unit] ),

    Designations,

    Donations[refunded] = FALSE (),

    NOT ISBLANK ( Donations[Giving Unit] )

)

Given Last 12 Months =

VAR asOf = [Report Date]

RETURN

    CALCULATE (

        [Total Given],

        REMOVEFILTERS ( 'Date' ),

        DATESBETWEEN ( 'Date'[Date], EDATE ( asOf, -12 ) + 1, asOf )

    )

Given Prior 12 Months =

VAR asOf = [Report Date]

RETURN

    CALCULATE (

        [Total Given],

        REMOVEFILTERS ( 'Date' ),

        DATESBETWEEN ( 'Date'[Date], EDATE ( asOf, -24 ) + 1, EDATE ( asOf, -12 ) )

    )

Giving Units Last 12 Months =

VAR asOf = [Report Date]

RETURN

    CALCULATE (



        [Giving Units],

        REMOVEFILTERS ( 'Date' ),

        DATESBETWEEN ( 'Date'[Date], EDATE ( asOf, -12 ) + 1, asOf )

    )

Giving Units Prior 12 Months =

VAR asOf = [Report Date]

RETURN

    CALCULATE (

        [Giving Units],

        REMOVEFILTERS ( 'Date' ),

        DATESBETWEEN ( 'Date'[Date], EDATE ( asOf, -24 ) + 1, EDATE ( asOf, -12 ) )

    )

Average per Giving Unit =

DIVIDE ( [Given Last 12 Months], [Giving Units Last 12 Months] )

Recurring Share General Fund =

DIVIDE (

    CALCULATE ( [General Fund Given], Donations[Is Recurring] = TRUE () ),

    [General Fund Given]

)

First-Time Givers =

VAR periodStart = MIN ( 'Date'[Date] )

VAR periodEnd = MAX ( 'Date'[Date] )

VAR unitsInPeriod =

    CALCULATETABLE (

        VALUES ( Donations[Giving Unit] ),

        Donations[refunded] = FALSE (),

        NOT ISBLANK ( Donations[Giving Unit] )

    )

RETURN

    COUNTROWS (

        FILTER (

            unitsInPeriod,

            VAR firstGift =

                CALCULATE (

                    MIN ( Donations[received_at] ),

                    REMOVEFILTERS ( 'Date' ),

                    Donations[refunded] = FALSE ()

                )

            RETURN firstGift >= periodStart && firstGift <= periodEnd

        )

    )

First-Time Givers Last 12 Months =

VAR asOf = [Report Date]

RETURN

    CALCULATE (



        [First-Time Givers],

        REMOVEFILTERS ( 'Date' ),

        DATESBETWEEN ( 'Date'[Date], EDATE ( asOf, -12 ) + 1, asOf )

    )

Lapsed Givers =

VAR asOf = [Report Date]

VAR lastYear = YEAR ( asOf ) - 1

VAR gaveLastYear =

    CALCULATETABLE (

        VALUES ( Donations[Giving Unit] ),

        REMOVEFILTERS ( 'Date' ),

        'Date'[Year] = lastYear,

        Donations[refunded] = FALSE (),

        NOT ISBLANK ( Donations[Giving Unit] )

    )

RETURN

    COUNTROWS (

        FILTER (

            gaveLastYear,

            CALCULATE (

                MAX ( Donations[received_at] ),

                REMOVEFILTERS ( 'Date' ),

                Donations[refunded] = FALSE ()

            ) < asOf - 90

        )

    )

Lapsed Givers Last Year Giving =

VAR asOf = [Report Date]

VAR lastYear = YEAR ( asOf ) - 1

VAR gaveLastYear =

    CALCULATETABLE (

        VALUES ( Donations[Giving Unit] ),

        REMOVEFILTERS ( 'Date' ),

        'Date'[Year] = lastYear,

        Donations[refunded] = FALSE (),

        NOT ISBLANK ( Donations[Giving Unit] )

    )

VAR lapsed =

    FILTER (

        gaveLastYear,

        CALCULATE ( MAX ( Donations[received_at] ), REMOVEFILTERS ( 'Date' ) ) < asOf - 90

    )

RETURN

    SUMX (

        lapsed,

        CALCULATE ( [Total Given], REMOVEFILTERS ( 'Date' ), 'Date'[Year] = lastYear )

    )



Units in Band =

VAR lo = SELECTEDVALUE ( 'Giving Bands'[Min] )

VAR hi = SELECTEDVALUE ( 'Giving Bands'[Max] )

VAR units =

    CALCULATETABLE (

        VALUES ( Donations[Giving Unit] ),

        Designations,

        Donations[refunded] = FALSE (),

        NOT ISBLANK ( Donations[Giving Unit] )

    )

RETURN

    COUNTROWS ( FILTER ( units, VAR t = [Total Given] RETURN t >= lo && t < hi ) )

Dollars in Band =

VAR lo = SELECTEDVALUE ( 'Giving Bands'[Min] )

VAR hi = SELECTEDVALUE ( 'Giving Bands'[Max] )

VAR units =

    CALCULATETABLE (

        VALUES ( Donations[Giving Unit] ),

        Designations,

        Donations[refunded] = FALSE (),

        NOT ISBLANK ( Donations[Giving Unit] )

    )

RETURN

    SUMX ( FILTER ( units, VAR t = [Total Given] RETURN t >= lo && t < hi ), [Total Given] )

Share of Units =

DIVIDE ( [Units in Band], [Giving Units] )

Share of Dollars =

DIVIDE ( [Dollars in Band], [Total Given] )

Budget =

SUM ( 'General Fund Budget'[budget_amount] )

Budget Difference =

[General Fund Given] - [Budget]

General Fund YTD =

CALCULATE ( [General Fund Given], DATESYTD ( 'Date'[Date] ) )

Budget YTD =

CALCULATE ( [Budget], DATESYTD ( 'Date'[Date] ) )

General Fund YTD Last Year =

CALCULATE ( [General Fund YTD], SAMEPERIODLASTYEAR ( 'Date'[Date] ) )



Budget Through Last Full Month =

VAR endM = EOMONTH ( [Report Date], -1 )

RETURN

    CALCULATE (

        [Budget],

        REMOVEFILTERS ( 'Date' ),

        DATESBETWEEN ( 'Date'[Date], DATE ( YEAR ( endM ), 1, 1 ), endM )

    )

General Fund Through Last Full Month =

VAR endM = EOMONTH ( [Report Date], -1 )

RETURN

    CALCULATE (

        [General Fund Given],

        REMOVEFILTERS ( 'Date' ),

        DATESBETWEEN ( 'Date'[Date], DATE ( YEAR ( endM ), 1, 1 ), endM )

    )

Percent of Budget to Date =

DIVIDE ( [General Fund Through Last Full Month], [Budget Through Last Full Month] )

Campaign Goal =

SUM ( 'Pledge Campaigns'[goal_cents] ) / 100

Pledged =

SUM ( Pledges[amount_cents] ) / 100

Pledge Count =

COUNTROWS ( Pledges )

Pledged Percent of Goal =

DIVIDE ( [Pledged], [Campaign Goal] )

Campaign Received =

CALCULATE ( [Total Given], Funds[name] = "Room to Grow" )

Campaign Received to Date =

VAR d = MAX ( 'Date'[Date] )

RETURN

    CALCULATE (

        [Total Given],

        Funds[name] = "Room to Grow",

        REMOVEFILTERS ( 'Date' ),

        'Date'[Date] <= d

    )



Pledge Pace =

VAR d = MAX ( 'Date'[Date] )

VAR monthsIn = DATEDIFF ( DATE ( 2025, 10, 1 ), d, MONTH ) + 1

RETURN

    CALCULATE ( [Pledged], REMOVEFILTERS ( 'Date' ) ) * MIN ( 1, MAX ( 0, monthsIn ) / 36 )

Expected to Date (all pledges) =

SUM ( Pledges[Expected to Date] )

The HTML dashboard spreads the current month's budget using last year's daily giving pattern, because recurring
gifts arrive on the 1st and 15th. The two "Through Last Full Month" measures are the simpler Power BI version: they
compare complete months only, which avoids a misleading partial month.

Step 4. Pages and visuals

Page size 16:9. Apply the theme first, then add a campus slicer on every page and sync it across pages
(View, Sync slicers).

Page: Overview

Visual Fields and measures Notes

Slicer Donations[campus] Vertical list on the left, single select
with an All option.

Five cards (or one
multi-row card)

Given Last 12 Months, Giving Units Last 12
Months, Recurring Share General Fund, First-
Time Givers Last 12 Months, Lapsed Givers

Put the prior-period measures in the
card's subtitle or a small text box
underneath.

Clustered column
chart with a line

X: Date[Month Start] (last 24 months).
Columns: General Fund Given. Line: Budget

Use a line and clustered column chart.
Both measures share one axis, so there
is no second scale.

Table Funds[name], Given Last 12 Months, Given Prior
12 Months

Add data bars to the first amount
column.

Clustered bar chart Y: Giving Bands[Band] (sorted by Sort). Values:
Share of Units, Share of Dollars

Apply a page filter for the last 12 months
on Date[Date].

Bar chart Y: Donations[payment_method]. X: Total
Given, shown as percent of grand total

Sort descending.

Page: Givers

Visual Fields and measures Notes

Column chart X: Date[Month Start]. Y: First-Time Givers Last 24 months.

Line chart X: Date[Month Start]. Y: Recurring Share
General Fund

Format the axis as a percentage from 0
to 100.



Visual Fields and measures Notes

Table Donations[Giving Unit] (or People names), last
gift date, Total Given for last year

Filter the table visual with Lapsed Givers
> 0 at the unit level, or build it from a
calculated table if the church prefers a
static list each month.

Matrix Giving Bands[Band], Units in Band, Share of
Units, Dollars in Band, Share of Dollars

Data bars on Share of Dollars.

Page: Funds and budget

Visual Fields and measures Notes

Line chart X: Date[Month Start] for the current year.
Lines: General Fund YTD, Budget YTD, General
Fund YTD Last Year

Three lines on one axis.

Matrix Rows: Date[Month]. Values: General Fund
Given, Budget, Budget Difference, and last
year's amount

Conditional font color on Budget
Difference (green above zero, red
below).

Column chart X: Date[Month]. Y: Budget Difference Conditional fill color by sign.

Page: Room to Grow campaign

Visual Fields and measures Notes

Five cards Campaign Goal, Pledged (with Pledge Count),
Campaign Received, Expected to Date (all
pledges), gifts from non-pledgers

Gifts from non-pledgers equal
Campaign Received minus the sum of
Pledges[Received].

Line and clustered
column chart

X: Date[Month Start] from September 2025.
Columns: Campaign Received. Lines:
Campaign Received to Date, Pledge Pace

One shared axis.

Table Pledges[Pace Status], Pledge Count, Pledged,
sum of Pledges[Received]

Sort by a helper column so the order
reads Ahead, On pace, Behind, Not
started.

Table Pledge size bands (add a calculated column on
Pledges with the same banding approach),
Pledge Count, Pledged

Optional.

Step 5. Check the numbers

With all campuses selected, the cards should match the HTML dashboard: giving units over the last 12
months, the General Fund recurring share, first-time and lapsed givers, and the Room to Grow pledged
and received totals. If a count is off, the usual cause is an ID column loaded as a number or the refunded
column loaded as text.



Sample data. All names, organizations, and figures are fictional.


